

From chaos to beauty

From the depths of scientific mystery and a basic knowledge of mathematics, weird and wonderful designs can be produced, from a computer with graphical output. Barry Martin and Mike Mudge present a kaleidoscope of patterns and figures.

'Chaos' is a respectable scientific discipline emerging in many areas, ranging from turbulent flow through population dynamics, heart cell dynamics and neurological patterns to the study of lasers. Any readers interested in the underlying theory can refer to the publication, *The Universality of Chaos*, a collection of papers edited by Predrag Cvitanovic and published by Adam Hilger (1984). However, with no knowledge of the theory and only a minimal knowledge of basic mathematics, patterns and designs of breathtaking beauty and intricacy can be produced — all that is required is a microcomputer with graphical output. Colour graphics, although not essential, can be used to produce subtle effects

and attractive designs.

The procedure is based on iterations of ordinary real numbers. A starting point (x_0, y_0) is chosen and then successive points (x_1, y_1) , (x_2, y_2) , and so on, are generated using some mathematical function. A digital computer plots the points so quickly that it appears as if electronic raindrops are falling onto the graphics screen. Eventually, after a thousand or so points have been plotted, a pattern begins to emerge. Patterns can be varied by changing the starting point or by changing numerical parameters in the mathematical function, thus giving infinite variety from any one point generator.

Do not, however, get the impression that the formulae have to be

complicated in order to produce exotic patterns. Very simple formulae using, for example, functions such as $\text{SGN}(x)$ and $\text{SQR}(\text{ABS}(x))$, which are available on all micros, can yield exquisite designs as can be seen from the detailed examples which appear in this article.

Some patterns seem to build up in one area of the screen for thousands of iterations and then, suddenly, the whole process jumps to another region and a further segment of the pattern develops. Obviously, some experimentation with scaling is usually necessary to reveal the full content of a given point generator. Increasing the magnification of an area surrounding a particular point of a pattern will reveal more of the




Fig 1




Fig 2




Fig 3




Fig 4

microstructure of the pattern and will probably produce a thing of beauty in itself.

Colours may be used to enhance the artistic appeal of the patterns, for example, by assigning colour according to how far the point being plotted is from the origin of coordinates. Alternatively, you can change colour according to the number of iterations required to compute the position of the point — for example, vary the colour every 100 iterations.

The techniques for developing the patterns described in this article were inspired by the Mandelbrot sets presented by Dr MM Novak and Jack Weber in the article 'Fractal Sets', PCW, December 1986. Mandelbrot's patterns emerge from complex numbers (not easy things for many people to manipulate on their computers) and the seeds for the iterative process are the points, infinite in number, found in a region of the plane. For every point on the screen a complicated calculation needs to be done many times over, and the computer generation of a complete pattern may take hours or even days, although Weber does describe ways to reduce this computation time.

Furthermore, to obtain detailed resolution of the final picture, high-powered graphics facilities are required. The process described here, on the other hand, uses only real numbers and the pattern grows from only a single seed, the starting point (x_0, y_0). The pattern begins to appear after a minute or so of computation, and exciting pictures can be produced using nothing more elaborate than the screen of a portable TV set.

Using such pattern-generating techniques, it is possible to design your own personalised wallpaper or

textiles so that no-one else in the world would be likely to have the same pattern. Imagine going into a shop, keying a couple of your own secret numbers into a computer terminal, and a computer-based automatic 'pattern designer' then produces rolls of unique personalised wallpaper or even a unique dress fabric.

However, to return to the 'chaos' of reality upon which all of this is based, before you all branch out into computerised wallpaper design, we will look at a one-dimensional case (no pretty pictures) which will illus-

trate how chaotic phenomena rise from an ordered situation.

One-dimensional chaos

As an example we will look at the equation (known to mathematicians as the logistic equation) (1) $\dots x_r + 1 = 4Cx_r(1 - x_r)$, where $r = 0, 1, 2, \dots$ successively. We take a starting value x_0 anywhere between 0 and 1 and then compute x_1, x_2, x_3, \dots using equation (1).

A procedure such as this is called an 'iterative scheme' and the computation of each successive x — value is called an iteration. Before

```

10 REM PROGRAM "HOPALONG".
20 REM THIS GENERATES PATTERNS BY ITERATION FROM "HOPALONG".
40
50 MODEL
60 REM READ IN DATA: CONSTANTS P,Q,R, INITIAL POINT, 'CENTRE' POSITION,
70 REM SCALE AND NUMBER OF ITERATIONS.
80 INPUT "CONSTANT P ",P
90 INPUT "CONSTANT Q ",Q
100 INPUT "CONSTANT R ",R
110 INPUT "X COORDINATE OF STARTING POINT ",X
120 INPUT "Y COORDINATE OF STARTING POINT ",Y
130 INPUT "X COORDINATES OF CENTRE POINT ",XC
140 INPUT "Y COORDINATES OF CENTRE POINT ",YC
150 INPUT "HORIZONTAL SCALE ",XS
160 INPUT "VERTICAL SCALE ",YS
170 INPUT "NUMBER OF ITERATIONS REQUIRED ",N
180 CLS
190 ICOUNT=0
200 T=1/SQR(2)
210
220 REM START ITERATION.
230 FOR I=1 TO N
240 X1=Y-SGN(X)*SQR(ABS(Q*X-R))
250 Y=P-X
260 X=X1
270 ICOUNT=ICOUNT+1
280 PRINT TAB(0,0);ICOUNT
290 REM ROTATE PLOT THROUGH -45 DEGREES.
300 U=T*X+T*Y
310 V=-T*X+T*Y
320 PLOT 69,XC+XS*U,YC+YS*V
330 NEXT I
340 STOP

```


GRAPHICS

starting, however, we must decide on a value for the constant C , as the behaviour of the iterative scheme is dependent on the constant we choose. Firstly, C must lie between 0 and 1; this keeps the successive iterates within the range zero to one. If C is less than 0.25, the values of x_r as r becomes large approach zero. If C lies between 0.25 and 0.75, the values of x_r approach $1 - 1/4C$.

For C between 0.75 and 0.9, more interesting features begin to emerge. For example, when $C=0.76$, x_r alternately approaches the two values of 0.589356083 ... and 0.730591286 ... and, as larger values of C less than 0.9 (or, more accurately, 0.8925 ...) are chosen, x_r simultaneously approaches an increasing number of these 'limit' values, as shown in the box below.

C	Number of 'limit' values
0.2	1
0.24	1
0.76	2
0.87	4
0.89	8
0.892	16

Limit values

The algebraic significance of this phenomenon is well-understood by the mathematical pundits and is, indeed, quite predictable and orderly. However, as C gets closer to the special value 0.8925 ..., order ceases and chaos takes over. The values of x_r become distributed between 0 and 1 in a thoroughly haphazard way, but order can return as C is further increased towards unity. For example, if you try $X_0=0.96$, you should find that just three points are visited.

```

10 REM PROGRAM "AXHEADS".
20 REM THIS GENERATES ITERATES WHICH "TILE" THE PLANE
30 REM WITH AXEHED LIKE FORMS.
40
50
60 MODEL
70 REM READ IN DATA: CONSTANT A, INITIAL POINT, 'CENTRE' POSITION,
80 REM SCALE AND NUMBER OF ITERATIONS.
90 INPUT "CONSTANT A ",A
100 INPUT "X COORDINATE OF STARTING POINT ",X
110 INPUT "Y COORDINATE OF STARTING POINT ",Y
120 INPUT "X COORDINATES OF CENTRE POINT ",XC
130 INPUT "Y COORDINATES OF CENTRE POINT ",YC
140 INPUT "HORIZONTAL SCALE ",XS
150 INPUT "VERTICAL SCALE ",YS
160 INPUT "NUMBER OF ITERATIONS REQUIRED ",N
170 CLS
180 ICOUNT=0
190 T=1/SQR(2)
200
210 REM START ITERATION.
220 FOR I=1 TO N
230 X1=Y-SIN(X)
240 Y=A-X
250 X=X1
260 ICOUNT=ICOUNT+1
270 PRINT TAB(0,0);ICOUNT
280 REM ROTATE PLOT THROUGH -45 DEGREES.
290 U=T*X+T*Y
300 V=-T*X+T*Y
310 PLOT 69,XC+XS*U,YC+YS*V
320 NEXT I
330 STOP

```

All the various features described above can be displayed using the program 'CHAOSMW' on page 122, which is written in BBC Basic. This first draws the parabola having equation (2) $y=4Cx(1-x)$, together with the straight line having equation (3) $y=x$. It then carries out the iterative process which is illustrated in Fig 1 where, in this case, $C=0.8$ and $x_0=0.3$.

CHAOSMW originated from Matthew Wells, who produced it as part of an undergraduate project at the University of Aston, Birmingham

in 1986. However, he is making no claims to be the first to write such a program: many mathematicians during the last few years have constructed similar routines.

To help you centre and scale the plots on your own monitor, it will be useful to know that on a screen of 1280 X 1024 pixels, $x_0 = 200$, $y_0 = 100$, $x_s = 900$ and $y_s = 900$ gives a good picture.

Two-dimensional chaos

The two-dimensional mappings which we will consider are of the




Fig 5




Fig 6

form (4) $x_r+1=f(x_r, y_r)$, $y_r+1=A-x_r$, where A is a numerical constant, $r = 0, 1, 2 \dots$ successively and starting values (x_0, y_0) are to be specified, and $f(x, y)$ is a non-linear function of the ordered pair (x, y) . Readers have the opportunity to investigate the consequences of replacing $y_r+1=A-x_r$ by $y_r+1=g(x_r, y_r)$, a second non-linear function. Brief experimentation within this area, however, suggests that the resulting patterns may be over-elaborate with a consequent loss of aesthetic appeal.

The first family of mappings to be considered is (5) $x_r+1=y_r-\sin(x_r)$,

$y_r+1=A-x_r$, where $r=0, 1, 2 \dots$ successively with starting values of $(0, 0)$. The necessary program 'AXHEADS', is listed on page 120.

It seems that if $A=\pi=3.1415026 \dots$, this program displays a repetition of the four points $(0, 0)$, $(0, \pi)$, (π, π) and $(\pi, 0)$. This phenomenon is somewhat analogous to the predictable and orderly behaviour previously described for the one-dimensional mapping equation (1) when the constant C is less than $0.87 \dots$. We are not, therefore, surprised by the approximation to four-fold symmetry that results from choosing a value for A which is sufficiently close to π

(see Fig 2 referring to $A=3.1421$).

20,000+ points have been plotted and the colour changed every 50 iterations. However, the results associated with the value $A=3.15$ and using different numbers of iterations are displayed in Figs 3 and 4. These pictures display some of the infinite variety associated with the point-generator equation (5), as discussed in the introduction.

Scaling with $A=3.12$, and initial point $(0, 0)$ on the screen mentioned above, is reasonable with $x_c=600$, $y_c=550$, $x_s=10$ and $y_s=10$.

A second family of mappings to be considered is (6) $x_r+1=y_r-\text{sign}(x_r) \times |Qx_r-R|$, $y_r+1=P-x_r$, where P , Q and R are numerical constants, $r=0, 1, 2 \dots$ successively and starting values (x_0, y_0) are to be specified.

The case when $x_0=0$, $y_0=0$, $R=0$ and $Q=1$ has been discussed by Martin, *The Mathematical Gazette*, volume 70, number 452, June 1986, pages 140-142, for a range of values of P .

The general investigation of the point generator, equation (6), may be carried out using the program 'HOPALONG' on page 119 of this article (the name was coined by AK Dewdney who writes on Computer Recreations in *Scientific American* magazine) for which specimen output is provided in the form of Figs 5 and 6.

History

This approach to pattern generation has been inspired by the analysis of the so-called Hénon Attractor defined by (7) $x_r+1=1+y_r-ax_r^2$, $y_r+1=bx_r$, where a and b are numerical constants, $r=0, 1, 2 \dots$ successively and starting values (x_0, y_0) are to be specified (see, for example, M Hénon, *Communications in Mathematical Physics*, volume 50, pages 69-77, 1976). The chaotic situation resulting when $a=1.4$ and $b=0.3$ is displayed through four stages of magnification using up to 5×10^6 points with starting values $(0, 0)$. The black-on-white pictures presented have no artistic merit, but did prompt considerable mathematical investigations resulting in part from their apparent 'ultimate' simplicity.

Such advantages seem unlikely in relation to the point generators (5) and (6) featured here, although some mathematicians are optimistic. Readers are encouraged to experiment with alternative forms of (4) — that is, to develop new designs rather than to seek underlying advances in the rather difficult mathematical theory.

Barry Martin is a lecturer in mathematics at Aston University, Birmingham. Mike Mudge is the author of PCW's regular maths column, Numbers Count (page 211).

END

```

10 REM PROGRAM "CHAOSMW".
20 REM THIS PROGRAM DISPLAYS SUCCESSIVE ITERATES OF THE LOGISTIC EQUATION
30 REM X(N+1)=4*C*X(N)*(1-X(N)).
40 REM TO START IT ASKS FOR A VALUE OF C BETWEEN 0 AND 1
50 REM AND FOR AN INITIAL VALUE OF X BETWEEN 0 AND 1.
60 REM THIS PROGRAM WAS WRITTEN BY MATTHEW WELLS. 1986.
70 MODE1
80
90 INPUT "CONSTANT 0<C<1 ",C
100 INPUT "STARTING VALUE 0<STARTX<1 ",STARTX
110 INPUT "MAXIMUM NUMBER OF ITERATIONS ",MAXIT
120 INPUT "X COORDINATE OF ORIGIN ",XO
130 INPUT "Y COORDINATE OF ORIGIN ",YO
140 INPUT "LENGTH IN PIXELS OF X-AXIS ",XS
150 INPUT "LENGTH IN PIXELS OF Y-AXIS ",YS
160
170 REM DRAW AXES AND THE PARABOLA 4*C*X*(1-X).
180 CLS
190 PRINT TAB(0,0)
200 PRINT TAB(4,31) "0"
210 PRINT TAB(0,0)
220 PRINT TAB(34,31) "1"
230 PRINT TAB(0,0)
240 PRINT "C=";C
250 PRINT "X(0)=";STARTX
260 PLOT 4,XO,YO
270 PLOT 5,XO+XS,YO
280 PLOT 4,XO,YO
290 PLOT 5,XO,YS
300 PLOT 4,XO,YO
310 PLOT 5,XO+XS,YO+YS
320 X=0.0
330 SCALE=1/YS
340 FOR I=1 TO YS
350 X=X+SCALE
360 PLOT 69,XO+X*XS,YO+YS*4*X*C*(1-X)
370 NEXT I
380
390 M=0
400 X=STARTX
410 PLOT 4,XO+STARTX*XS,YO
420 GCOL 0,1
430 REM START ITERATION.
440 Y=4*X*C*(1-X)
450
460 REM PAUSE TO SEE DEVELOPMENT OF PLOT.
470 FOR N=0 TO 400
480 NEXT
490 PLOT 5,XO+X*XS,YO+YS*Y
500 PLOT 5,XO+Y*XS,YO+Y*YS
510 X=Y
520 M=M+1
530 REM INTRODUCE COLOUR CHANGE AFTER 50 ITERATIONS.
540 IF M<50 THEN GOTO 560
550 GCOL 0,2
560 IF M=MAXIT THEN STOP
570 GOTO 440

```